
Krzysztof Jankiewicz 

PL/SQL – co, gdzie i dlaczego? 

PL/SQL to język proceduralny rozszerzający deklaratywne możliwości języka SQL o proceduralne konstrukcje takie 

jak: zmienne, pętle, instrukcje warunkowe, obsługę wyjątków. 

O ile SQL pozwala na tworzenie zapytań (np. SELECT) lub poleceń (np. CREATE, DELETE itp.) o tyle PL/SQL pozwala 

na tworzenie programów. 

Program PL/SQL, który może być wykonywany przez bazę danych w oparciu o treść PL/SQL, która 

 napisana zostanie i wysłana z poziomu narzędzia/aplikacji (np. SQL Developer/APEX) 

 jest składowana na poziomie bazy danych – podobnie jak treść SQL może być składowana na poziomie bazy 

danych pod postacią perspektyw (ang. view), tak treść PL/SQL może być składowana pod postacią funkcji, 

procedur, pakietów, wyzwalaczy, definicji typów obiektowych i pewnie jeszcze kilku innych konstrukcji 

 

Aby poznać podstawy PL/SQL i zrozumieć powyższe stwierdzenia wykonaj poniższe zadania. 

Zadanie 1 – uruchomienie kodu PL/SQL z narzędzia 

Zaloguj się do bazy danych za pomocą SQL Developera. W oknie, w którym zazwyczaj wprowadzasz polecenia SQL, 

tym razem wprowadź następujący kod PL/SQL. 

SET SERVEROUTPUT ON 

declare 

  v_placa_max number; 

  cursor c_zesp is  

    select * from zespoly; 

begin 

  for r_zesp in c_zesp loop 

    select max(placa_pod)  

    into   v_placa_max 

    from   pracownicy  

    where  id_zesp = r_zesp.id_zesp; 

    dbms_output.put_line('W zespole '||r_zesp.NAZWA|| 

                         ' największa placa to: '||v_placa_max); 

  end loop;   

end; 

 

Następnie stań kursorem na początku kodu i uruchom go za pomocą klawisza F5 lub przycisku jak na poniższym 

rysunku. 

Wynik powinien być następujący: 

 

Co się stało? Napisaliśmy kod PL/SQL w narzędziu SQL Developer, a następnie wysłaliśmy go do bazy danych celem 

jego wykonania.  

Zadanie 2 – wykorzystanie kodu PL/SQL składowego w bazie danych 

Kod PL/SQL może też znajdować się w bazie danych. Może tam być umieszczony przez programistów, np. przez 

Ciebie, kiedy będzie Ci to potrzebne i będzie to w zakresie Twoich umiejętności. Jednak od razu, bezpośrednio po 

instalacji bazy danych, w bazie danych znajduje się wiele kodu PL/SQL, mającego głównie postać tzw. pakietów 

wbudowanych, który może być bardzo zgrabnie wykorzystywany do najróżniejszych celów. 

Przykładowo w kodzie PL/SQL, który przed chwilą wysłaliśmy do bazy danych, we fragmencie 

dbms_output.put_line('W zespole '||r_zesp.NAZWA|| 

                         ' największa placa to: '||v_placa_max); 

użyliśmy pakietu wbudowanego dbms_output, aby wyświetlić na ekranie pewne informacje.  

Poniżej skorzystamy z innego pakietu - dbms_metadata.  Użyjemy go do wygenerowania polecenia DDL 

tworzącego tabelę PRACOWNICY.  

Wprowadź tym razem jako treść polecenia  

SET LONG 20000 

SET PAGESIZE 10000 

select dbms_metadata.get_ddl('TABLE', 'PRACOWNICY', USER)  

from dual; 

I ponownie uruchom je za pomocą przycisku F5. Tym razem wynik powinien być nieco inny: 

 

Czy to nie przydatne? Łatwo wydobyć za pomocą tego pakietu (dbms_metadata) definicje różnych obiektów w 

bazie danych i utworzyć, za ich pomocą, kopie identycznych obiektów w innej bazie danych. Czy byłoby możliwe 

napisanie „czystego” deklaratywnego polecenia SQL, który wygenerowałoby kod DDL dla wskazanej tabeli? Być może 

tak, jednak miałoby to polecenie wiele linii kodu i byłoby z pewnością bardzo trudne i niezrozumiałe. Tymczasem ta 

sama czynność za pomocą proceduralnego języka PL/SQL, to stosunkowo proste zadanie. Język proceduralny potrafi 

uzyskiwać znacznie więcej i w często znacznie prostszy i przejrzystszy sposób. 

Inne wbudowane pakiety pozwalają na jeszcze ciekawsze rzeczy np. zbieranie statystyk (celem wydajniejszej pracy 

bazy danych), przetwarzanie danych przestrzennych czy obrazów, wysyłanie maili i wiele innych.  

Znając PL/SQL możesz robić wszystkie te rzeczy – programować wydajne i funkcjonalne aplikacje bazodanowe.  

 

 

 

Tymczasem zaczniemy od czegoś prostszego. Spróbujemy napisać procedurę przeznaczoną do zatrudniania 

pracowników zgodnie z regułami biznesowymi działającymi w „naszej” firmie.  

Zatrudnienie pracownika NOWAK w zespole ADMINISTRACJA na stanowisku ASYSTENT z pensją 1500, którego 

szefem będzie WEGLARZ, powinno skutkować wykonaniem następujących czynności: 

a) Pobierz identyfikator dla nowo zatrudnionego pracownika 

b) Pobierz identyfikator zespołu ADMINISTRACJA 

c) Sprawdź czy pensja pasuje do widełek asystentów 

d) Pobierz identyfikator Węglarza – szefa nowego pracownika 

e) Jeśli wszystko jest w porządku wprowadź do bazy danych nowego pracownika, w przeciwnym przypadku 

poinformuj użytkownika procedury co zrobił nie tak 

Standardowo, przy wykorzystaniu „czystego” SQL nasze zadanie zatrudnienia pracownika wiązałoby się z 

wykonaniem wielu poleceń SQL, odbierania wyników, zapisywania ich „na kartce”, wykorzystywania tych wyników w 

kolejnych poleceniach, porównywania zapisywanych wartości, podejmowania na podstawie porównań różnych 

decyzji itd.  

My chcemy to zrobić za pomocą jednego polecenia – polecenia PL/SQL, które zostanie wysłane raz, wykona 

wszystkie zadania po stronie bazy danych, a następnie „odda” nam wynik. Na początku polecenie to (cały kod 

PL/SQL) będzie wysyłane z poziomu SQL Developera, później umieścimy ten kod w bazie danych, aby móc go 

następnie wywoływać w o wiele prostszy sposób. 

Zadanie 3 – blok PL/SQL 

Zanim zaczniemy, musimy wprowadzić kilka podstaw. Kod większości proceduralnych języków programowania 

umieszczany jest zawsze w blokach programowych, które w PL/SQL zaczynają się od słowa BEGIN i kończą na słowie 

END.  

Każda instrukcja w PL/SQL kończona jest za pomocą średnika. Ze względu na to, że blok PL/SQL można potraktować 

jako złożoną instrukcję, zawierającą instrukcje składowe, to słowo END także kończymy średnikiem. 

Skasuj zawartość treści polecenia w SQL Developerze i napisz zatem  

 

Blok PL/SQL nigdy nie może być pusty – musi zawierać instrukcje,  

więc tymczasowo wpisz pomiędzy BEGIN i END  

instrukcję pustą NULL;  Pamiętaj o średniku. 

Wygląd Twojego kodu oraz efekt po  

uruchomieniu za pomocą F5, przedstawiony  

został obok.  

 

Zadanie 4 – zmienna i SELECT INTO 

Wróćmy do naszego zadania zatrudnienia pracownika.  

Na początek podpunkt a) – pobranie identyfikatora dla nowo zatrudnionego pracownika. Standardowo powinniśmy 

wykorzystać do tego celu sekwencje. My zrobimy to trochę mniej zgrabnie – obliczymy maksymalną wartość 

wykorzystywaną przez identyfikatory pracowników i dodamy do niej 10. Polecenie, które to potrafi ma postać 

select max(id_prac)+10 from pracownicy; 

Niestety bezpośrednie wykorzystanie tego polecenia w PL/SQL nie wchodzi w grę. Po pierwsze w PL/SQL nie można 

korzystać z poleceń SELECT w formie znanej nam do tej pory – zamiast nich dostępne są polecenia SELECT … 

INTO …, których zadaniem jest pobrać wynik zapytania do zmiennych. A po drugie, no właśnie, nie mamy jeszcze 

zmiennych.  

Zatem zaczniemy od zmiennych. Zmienne w PL/SQL pozwalają przechowywać wartości podczas przetwarzania 

poleceń zawartych w bloku PL/SQL. Aby móc z nich skorzystać trzeba je zadeklarować.  

Deklaracja zmiennych dostępna jest w części deklaracyjnej bloku PL/SQL, która rozpoczyna się przed słowem BEGIN 

za pomocą słowa DECLARE.  

Przyglądnij się przykładowi w dokumentacji 

https://docs.oracle.com/cd/E11882_01/appdev.112/e25519/fundamentals.htm#LNPLS208 

Do przechowania identyfikatora potrzebujemy zmiennej numerycznej, 

a zatem zadeklarujemy ją w następujący sposób:  

  

https://docs.oracle.com/cd/E11882_01/appdev.112/e25519/fundamentals.htm#LNPLS208

A następnie dodamy nasze polecenie SELECT uzupełnione o klauzulę INTO. 

Wstaw do wnętrza bloku PL/SQL zamiast instrukcji NULL; poniższe polecenie: 

select max(id_prac)+10  

into v_new_id_prac 

from pracownicy; 

Wnętrze bloku PL/SQL, ta część pomiędzy BEGIN i END; to część instrukcji, to tam umieszczamy wszelkie instrukcje 

i polecenia naszego kodu PL/SQL. 

Aby sprawdzić czy wszystko działa poprawnie, wstaw pod powyższym poleceniem kolejne polecenie, które wyświetli 

nam wartość zmiennej v_new_id_prac po wykonanej instrukcji  

SELECT … INTO … 

DBMS_OUTPUT.put_line(v_new_id_prac); 

Kompletny kod, oraz rezultat jego uruchomienia powinien być jak na zrzucie 

ekranu obok: 

 

Zadanie 5 – powtórka  

Czas na podpunkt b) – pobranie identyfikatora zespołu ADMINISTRACJA.  

Pracownik, którego chcemy zatrudnić ma zostać zatrudniony w zespole ADMINISTRACJA. Jednak aby wstawić 

pracownika musimy znać identyfikator tego zespołu.  

Zapytanie, które rozwiązuje powyższy problem – pobiera identyfikator zespołu na podstawie jego nazwy może 

wyglądać następująco: 

select id_zesp from zespoly where nazwa = 'ADMINISTRACJA'; 

Wykorzystaj to zapytanie do pobrania do kolejnej zmiennej potrzebnego identyfikatora zespołu. Pamiętaj o 

deklaracji nowej zmiennej – niech jej nazwa to v_id_zesp. Pamiętaj także, że polecenia SELECT w PL/SQL muszą 

mieć formę SELECT … INTO …. 

Ponadto dodaj polecenie, które wyświetli uzyskany identyfikator zespołu. Aby rozpoznać, która z wyświetlanych liczb 

co oznacza, dodaj stosowną informację przed wartością zmiennej: 

DBMS_OUTPUT.put_line('Id zespołu: '||v_id_zesp); 

Uruchom otrzymany kod i sprawdź czy wynik jest zgodny z poniższym: 

Nowy id pracownika: 240 

Id zespołu: 10 

Oczywiście popraw także sposób wyświetlania identyfikatora pracownika.  

Jeśli masz kłopoty z uzyskaniem tego wyniku popatrz na poniższą podpowiedź. 

 

Zadanie 6 – komentarze  

Niektórzy uważają, że programy pisze się nie po to aby działały (choć to z pewnością ważny efekt uboczny), ale po to 

aby je czytać (w szczególności kiedy usiłujemy znaleźć błąd w sytuacji kiedy efekt uboczny nie występuje). W związku 

z tym ważna jest czytelność programów. W podniesieniu czytelności bez wątpienia służą komentarze – ciągi znaków, 

które nie są kodem (nie są wykonywane) a jedynie służą do opisu kodu znajdującego się w pobliżu.  

W PL/SQL istnieją dwa rodzaje komentarzy – jednoliniowe i wieloliniowe. My wykorzystamy teraz komentarz 

jednoliniowy, który zaczyna się dwoma myślnikami i ciągnie się do końca linii.  

Pierwszy komentarz umieścimy przed pierwszym poleceniem 

SELECT … INTO …. 

-- a) Pobierz nowy identyfikator   

Drugi komentarz o treści b) Pobierz identyfikator 

zespołu ADMINISTRACJA umieść tam gdzie powinien 

zostać umieszczony  

Więcej o komentarzach możesz przeczytać oczywiście w 

dokumentacji 

https://docs.oracle.com/cd/E11882_01/appdev.112/e25519/fundamentals.htm#LNPLS203 

  

https://docs.oracle.com/cd/E11882_01/appdev.112/e25519/fundamentals.htm#LNPLS203

Zadanie 7 – powtórka i instrukcja warunkowa  

Czas na podpunkt c) Sprawdź czy pensja pasuje do widełek asystentów. W tym celu: 

 Zadeklaruj dwie zmienne numeryczne v_placa_min i v_placa_max 

 Wstaw w odpowiednim miejscu w bloku instrukcji (najlepiej tuż przed końcem bloku PL/SQL) odpowiedni 

komentarz 

 Po komentarzu wstaw polecenie SELECT … INTO …, które pobierze płace minimalne oraz maksymalne 

dla etatu ASYSTENT.  

Uwaga: polecenie może pobrać wartości dwóch kolumn do dwóch zmiennych za jednym razem.   

 Po pobraniu wartości zmiennych wyświetl ich wartości. 

Oczekiwany rezultat (oczywiście po uruchomieniu napisanego kodu): 

Nowy id pracownika: 240 

Id zespołu: 10 

Placa min: 300, placa max: 500 

 

Zaczyna być nudno? Trzeci raz to samo… Czas zatem na nowe elementy. Naszym zamiarem jest wstawić nowego 

pracownika na etacie ASYSTENTA z płacą 1500. Nasz kod ma sam sprawdzić czy te wartości są akceptowalne – czy 

proponowana płaca jest zgodna z widełkami na tym etacie. Aby to mógł zrobić musimy skorzystać z instrukcji 

warunkowej. Jest ich kilka w PL/SQL, my wykorzystamy najprostszą – IF.  

W tym celu poniżej polecenia wyświetlającego wartości zmiennych v_placa_min i v_placa_max napisz 

następujący fragment  

IF () THEN 

ELSE 

END IF; 

W nawiasie programista musi wstawić warunek, który ma być sprawdzany. Pomiędzy THEN a ELSE programista 

wstawia instrukcje, które mają być wykonane gdy warunek jest spełniony. Pomiędzy ELSE a END IF; programista 

wstawia instrukcje, które mają być wykonane w przeciwnym przypadku (czyli wówczas gdy warunek spełniony nie 

będzie.  

Naszym warunkiem będzie  

1500 between v_placa_min and v_placa_max   

Umieść go w odpowiednim miejscu. Na razie instrukcjami w obu przypadkach niech będą instrukcje NULL; 

Ostatecznie końcowy fragment kodu naszego programu powinien być następujący: 

 

Ta jak wspomnieliśmy istnieje wiele różnych i ciekawych instrukcji warunkowych w PL/SQL. Instrukcja warunkowa IF 

w postaci użytej powyżej to tylko jeden z przykładów. Więcej znajdziesz oczywiście w dokumentacji 

https://docs.oracle.com/cd/E11882_01/appdev.112/e25519/controlstatements.htm#LNPLS00402 

 

Zadanie 8 – zgłaszanie błędów 

Uzupełnienie nasze instrukcji warunkowej zaczniemy od instrukcji odpowiadającej za przypadek, gdy płaca nie 

spełnia reguł narzuconych przez widełki etatowe. Chcemy aby użytkownik dowiedział się o problemie, tak samo jak 

dowiaduje się kiedy narusza ograniczenia integralnościowe, czyli za pomocą błędu.  

Do generowania błędów służy specjalna funkcja RAISE_APPLICATION_ERROR(). Posiada ona dwa argumenty, 

numer błędu oraz komunikat błędu. Dla niesystemowych błędów (błędów generowanych przez programistę) numery 

muszą być nie większe niż -20000 i nie mniejsze niż  

-20999. Komunikat może być dowolny. Uwzględniając powyższe, umieść w części instrukcji dla przypadku, w którym 

warunek instrukcji warunkowej nie jest spełniony, następujący fragment kodu (obecną tam instrukcję NULL; 

możesz usunąć): 

RAISE_APPLICATION_ERROR(-20000, 'Płaca pracownika wykracza poza ustalone widełki'); 

Uruchom program. Tym razem rezultat nie jest optymistyczny. I taki być powinien dla płacy 1500. 
Error report - 

ORA-20000: Płaca pracownika wykracza poza ustalone widełki 

ORA-06512: przy linia 27 

20000. 00000 -  "%s" 

*Cause:    The stored procedure 'raise_application_error' 

           was called which causes this error to be generated. 

*Action:   Correct the problem as described in the error message or contact 

           the application administrator or DBA for more information. 

. . . 

  

https://docs.oracle.com/cd/E11882_01/appdev.112/e25519/controlstatements.htm#LNPLS00402

Zadanie 9 – prawie koniec 

Czas na podpunkt d) Pobierz identyfikator Węglarza – szefa nowego pracownika.  

W tym celu za chwilę wykonasz poniższe polecenia, jednak zanim je napiszesz zastanów się nad ich właściwym 

miejscem: 

 Zadeklaruj dwie zmienną numeryczną v_id_szefa 

 Wstaw w odpowiednim miejscu w bloku instrukcji odpowiedni komentarz 

 Wstaw polecenie SELECT … INTO …, które pobierze identyfikator pracownika o nazwisku WEGLARZ.  

 Wyświetl wartości pobranych zmiennych. 

Sprawdź czy wszystkie nowo  instrukcje (poza deklaracją zmiennych) umieszczone zostały w części instrukcji 

uruchamianej w przypadku, gdy warunek instrukcji warunkowej jest spełniony – obecną tam wcześniej instrukcję 

NULL; możesz usunąć. 

Końcowy fragment naszego programu powinien teraz wyglądać następująco: 

 

No i finał. Mamy już wszystko do wstawienia informacji o pracowniku. Brakuje tylko instrukcji INSERT. Zwróć 

uwagę, na prostotę użycia poleceń SQL w kodzie PL/SQL. To jedna z podstawowych zalet języka proceduralnego 

stworzonego do przetwarzania danych.  

Brakująca instrukcja będzie wykorzystywała wszystkie potrzebne zmienne i będzie wyglądała następująco: 

insert into pracownicy(id_prac, nazwisko, etat, placa_pod, id_zesp, id_szefa) 

    values (v_new_id_prac, 'NOWAK', 'ASYSTENT', 1500, v_id_zesp, v_id_szefa); 

Dodaj także instrukcję potwierdzającą wstawienie nowego pracownika.  

DBMS_OUTPUT.put_line('Nowy pracownik zostal wstawiony'); 

Z oczywistych względów nasz program w dalszym ciągu będzie zgłaszał błąd niepoprawnej płacy, dlatego przed jego 

uruchomieniem zmieńmy (w dwóch miejscach!) płacę, jaką ma otrzymać nasz NOWAK, na 400.  

Spróbuj teraz uruchomić prawie finalną wersję naszego programu.  

Rezultat powinien być następujący: 

Nowy id pracownika: 240 

Id zespołu: 10 

Placa min: 300, placa max: 500 

Id szefa: 100 

Nowy pracownik zostal wstawiony 

 

Jeśli taki nie jest, porównaj swój program z poniższym kodem.  

DECLARE 

  v_new_id_prac NUMBER; 

  v_id_zesp NUMBER; 

  v_placa_min NUMBER; 

  v_placa_max NUMBER; 

  v_id_szefa NUMBER; 

BEGIN 

  -- a) Pobierz nowy identyfikator  

  select max(id_prac)+10  

  into v_new_id_prac 

  from pracownicy; 

  DBMS_OUTPUT.put_line('Nowy id pracownika: '||v_new_id_prac); 

  -- b) Pobierz identyfikator zespołu ADMINISTRACJA 

  select id_zesp  

  into v_id_zesp  

  from zespoly  

  where nazwa = 'ADMINISTRACJA'; 

  DBMS_OUTPUT.put_line('Id zespołu: '||v_id_zesp); 

  -- c) Sprawdź czy pensja pasuje do widełek asystentów 

  select placa_min, placa_max  

  into   v_placa_min, v_placa_max  

  from   etaty 

  where  nazwa = 'ASYSTENT'; 

  DBMS_OUTPUT.put_line('Placa min: '||v_placa_min||', placa max: '||v_placa_max); 

  IF (400 between v_placa_min and v_placa_max) THEN 

    -- d) Pobierz identyfikator Węglarza – szefa nowego pracownika 

    select id_prac 

    into   v_id_szefa 

    from   pracownicy 

    where  nazwisko = 'WEGLARZ'; 

    DBMS_OUTPUT.put_line('Id szefa: '||v_id_szefa); 

    insert into pracownicy(id_prac, nazwisko, etat, placa_pod, id_zesp, id_szefa) 

    values (v_new_id_prac, 'NOWAK', 'ASYSTENT', 400, v_id_zesp, v_id_szefa); 

    DBMS_OUTPUT.put_line('Nowy pracownik zostal wstawiony'); 

  ELSE 

    RAISE_APPLICATION_ERROR(-20000, 'Płaca pracownika wykracza poza ustalone widełki'); 

  END IF; 

END; 

Zadanie 10 – duża różnica 

To co udało się nam do tej pory  zrealizować to oczywiście bardzo dużo.  

Zastanów się, które ze znanych Ci zalet korzystania z PL/SQL ma nasz kod w obecnej postaci.  

Niestety z praktycznego i użytkowego punktu widzenia nie osiągnęliśmy wiele. Nasz kod jest trudny w utrzymaniu i 

wykorzystywaniu. Każde zatrudnienie nowego pracownika wymaga modyfikacji wielu miejsc w posiadanym kodzie. 

Inne aplikacje, które łączą się z naszą bazą danych, będą musiały ten kod mieć u „siebie” – nie będą mogły naszego 

kodu współdzielić, itd. 

Aby to zmienić, utworzymy z posiadanego przez nas programu procedurę składowaną, której uruchamianie będzie o 

wiele łatwiejsze. 

W tym celu wykonamy kilka poniższych czynności. 

Zamiast słowa DECLARE wstaw tzw. nagłówek procedury o poniższej treści 

CREATE OR REPLACE PROCEDURE ZATRUDNIJ(P_NAZWISKO VARCHAR2, P_ETAT VARCHAR2,  

P_PLACA NUMBER, P_SZEF VARCHAR2, P_ZESPOL VARCHAR2) IS 

Dzięki temu nagłówkowi utworzymy procedurę ZATRUDNIJ, która będzie przyjmowała kilka parametrów: 

 P_NAZWISKO – nazwisko nowego pracownika 

 P_ETAT – etat, na którym nowy pracownik ma być zatrudniony  

 P_PLACA – jego płaca 

 P_SZEF – nazwisko szefa 

 P_ZESPOL – zespół, w którym ma być zatrudniony  

Samo zadeklarowanie nagłówka procedury i określenie jej parametrów nie wystarczy, należy wykorzystać te 

parametry w kodzie naszego programu. Poniżej wskazówki tych zmian: 

 'NOWAK' - P_NAZWISKO (1x) 

 'ASYSTENT' - P_ETAT (2x) 

 400 - P_PLACA (2x) 

 'WEGLARZ' - P_SZEF (1x) 

 'ADMINISTRACJA' - P_ZESPOL (1x) 

Jeśli wszystkie zmiany są już wprowadzone spróbuj wykonać teraz swój program 

Poprawny rezultat: 

Procedure ZATRUDNIJ compiled 

Co się stało, gdzie uzyskiwane poprzednio wyniki?  

No właśnie, tym razem nie uruchomiliśmy naszego programu.  

Dzięki poleceniu CREATE OR REPLACE umieściliśmy nasz program we wnętrzu bazy danych – utworzyliśmy 

procedurę składowaną. To pozwoli nam na wykorzystanie tego programu w o wiele prostszy sposób – pamiętasz jak 

w prosty sposób wygenerowaliśmy polecenie DDL tworzące tabelę PRACOWNICY? 

Spróbuj wykonać następujący fragment uruchamiający naszą procedurę z parametrami: 

BEGIN 

  ZATRUDNIJ('DRUGI NOWAK', 'PROFESOR', 400, 'NOWAK', 'ALGORYTMY'); 

END; 

 

Oczekiwany rezultat: 

ORA-20000: Płaca pracownika wykracza poza ustalone widełki 

ORA-06512: przy "KJANKIEWICZ.ZATRUDNIJ", linia 37 

ORA-06512: przy linia 2 

20000. 00000 -  "%s" 

*Cause:    The stored procedure 'raise_application_error' 

. . . 

No tak, płaca profesora powinna być nieco większa. Zmień płacę na wartość 1000 i spróbuj ponownie uruchomić 

Twój kod.  

Oczekiwany rezultat: 

Nowy id pracownika: 250 

Id zespołu: 40 

Placa min: 800, placa max: 1500 

Id szefa: 240 

Nowy pracownik zostal wstawiony  

Prawda że ten sam efekt (wstawienie nowego pracownika) został uzyskany za pomocą prostszego kodu PL/SQL?  

Zastanów się ponownie, które ze znanych Ci zalet korzystania z PL/SQL wykorzystywane są teraz. 

Teraz zrobiliśmy naprawdę wiele. Poznaliśmy podstawy PL/SQL, dowiedzieliśmy się jak stworzyć kod PL/SQL, jak 

umieścić go w bazie danych i jak go wykorzystywać. Dowiedzieliśmy się także jakie zalety może on ze sobą nieść. 

Oczywiście jest jeszcze wiele do zrobienia, aby osiągnąć poziom master. Nie poznaliśmy kursorów, nie poznaliśmy 

pętli, nie poznaliśmy innych instrukcji warunkowych. Nie utworzyliśmy funkcji, pakietu, wyzwalacza. Jednak mając 

podstawy możemy te braki uzupełnić w o wiele prostszy i bardziej zrozumiały sposób. Na zakończenie poznamy 

jeszcze jedną drobną, ale ważną rzecz – obsługę wyjątków. 

Zadanie 11 – obsługa wyjątków 

Nasz program potrafi generować błąd związany z płacą pracownika. Jest to błąd biznesowy, zrozumiały dla naszego 

użytkownika, ważny i spełniający pozytywną rolę. Niestety nasz program potrafi generować także błędy 

niezrozumiałe i nieoczekiwane. Aby się o tym przekonać wywołaj jeszcze raz tę samą procedurę w następujący 

sposób.  

BEGIN 

  ZATRUDNIJ('TRZECI NOWAK', 'REKTOR', 400, 'JANKIEWICZ', 'BAZY DANYCH'); 

END; 

Efekt powinien być następujący  

Error report - 

ORA-01403: nie znaleziono danych 

ORA-06512: przy "KJANKIEWICZ.ZATRUDNIJ", linia 15 

ORA-06512: przy linia 2 

01403. 00000 -  "no data found" 

*Cause:    No data was found from the objects. 

O co chodzi? Jakich danych nie znaleziono? Sprawa jest dość oczywista. Szefem trzeciego Nowaka miał być 

Jankiewicz, którego nie ma w bazie danych. Zespół, do którego chcieliśmy przypisać naszego pracownika to Bazy 

Danych – taki zespół też nie występuje. Etatem trzeciego Nowaka miał być Rektor. Rektor?! To tych danych nie 

znaleziono.  

A z technicznego punktu widzenia co się stało?  

Otóż polecenie SELECT … INTO … musi zwrócić dokładnie 1 wiersz (bo ma tylko jeden zestaw zmiennych do 

przypisania). Jeśli tak się nie stanie polecenie to zgłosi wyjątek  

 NO_DATA_FOUND – gdy zapytanie nie zwróci żadnego wyniku (nasz przypadek), lub  

 TOO_MANY_ROWS – gdy zapytanie zwróci więcej niż jeden wiersz.  

Jak zapobiec takim błędom – wyjątkom? Albo jak zmienić je na inne – biznesowe? Odpowiedź jest prosta obsłużyć  

je. Obsługa wyjątków może pojawić się w bloku PL/SQL przed słowem kluczowym END a słowem EXCEPTION, które 

rozpoczyna część obsługi wyjątków (to już trzecia poznana przez Ciebie część, z której może składać się blok PL/SQL). 

Postać obsługi wyjątków wygląda następująco: 

WHEN WYJĄTEK THEN instrukcje obsługi 

W naszym przypadku będziemy chcieli obsłużyć wyjątek NO_DATA_FOUND informując użytkownika, również za 

pomocą błędu, ale już biznesowego, że musi sprawdzić te trzy składowe – zespół, szefa i etat, bo któregoś z nich po 

prostu nie ma. Ostatecznie zatem nasza obsługa wyjątku może mieć postać: 

  WHEN NO_DATA_FOUND THEN  

    raise_application_error(-20001,'Błąd: Nie znaleziono Etatu, Szefa lub '|| 

                                   'Zespołu dla nowo zatrudnianego pracownika'); 

Tylko zaraz, gdzie to wpisać? Nasz kod PL/SQL jest teraz w bazie danych, a nie w SQL Developerze. Tu pozostało tylko 

wywołanie. Na szczęcie SQL Developer jak na narzędzie programistyczne przystało, pozwala także na edycję kodu 

PL/SQL zapisanego w bazie danych. 

W tym celu w nawigatorze połączeń (Connections), rozwiń gałąź Procedures i zaznacz 

procedurę ZATRUDNIJ.  

Otworzy się nowa zakładka z treścią zaznaczonej procedury. 

Przed końcowym słowem kluczowym END; w treści procedury, wstaw zatem słowo EXCEPTION (rozpoczynając w 

ten sposób część obsługi wyjątków) i wstaw pomiędzy te dwa słowa opracowaną powyżej obsługę wyjątków. 

Sprawdź, czy Twój kod wygląda podobnie jak ten poniżej. 

 

Aby zapisać wprowadzone zmiany wybierz przycisk Compile 

 

Jeśli nie ma żadnych błędów, to pozostaje sprawdzić czy rozwiązanie przynosi oczekiwany skutek.  

Powróć do miejsca wywołania naszej procedury i uruchom ją ponownie.  

BEGIN 

  ZATRUDNIJ('DRUGI NOWAK', 'PROFESOR', 400, 'JANKIEWICZ', 'BAZY DANYCH'); 

END; 

Rezultat… no cóż teraz wszystko (no prawie) wiadomo. 

Error report - 

ORA-20001: Błąd: Nie znaleziono Etatu, Szefa lub Zespołu dla nowo zatrudnianego pracownika 

ORA-06512: przy "KJANKIEWICZ.ZATRUDNIJ", linia 41 

ORA-06512: przy linia 2 

Teraz to już naprawdę koniec. Poznaliśmy wszystkie istotne konstrukcje PL/SQL. Jeśli zostały przez Ciebie 

zrozumiane, to dalsze pogłębianie wiedzy będzie zdecydowanie znacznie prostsze.  

A jest co pogłębiać…  

Kapitalnym źródłem wiedzy są: 

 dokumentacja  

 materiały na studium  

 różnorodne tutoriale i blogi dostępne w sieci 


